Desigualdades y funciones

Los objetivos de esta práctica son:

• Analizar diversas órdenes de Maple que afectan a igualdades y desigualdades.
• Definir funciones reales de una variable real y efectuar operaciones entre ellas.
• Dibujar gráficas sencillas.
• Hacer sencillos ejercicios de programación con los conceptos anteriores.

1. Funciones reales de una variable real. Desigualdades

1) Resolver la desigualdad $|3\,x-5|<3$.

Usamos la orden **solve**, con su sintaxis correspondiente, es decir, indicando cuál es la incógnita:

```
> solve(abs(3*x-5)<3,x);solve({abs(3*x-5)<3},x);
```

$$RealRange\left(Open\left(\frac{2}{3}\right), Open\left(\frac{8}{3}\right)\right)$$

$$\left\{\frac{2}{3}<x, x<\frac{8}{3}\right\}$$

Es el intervalo abierto (2/3,8/3).

2) Hallar los números que verifican simultáneamente las desigualdades: $\frac{x-1}{3\,x}<1$, $\frac{|x|+2}{x}<1$.

Con la orden **solve** se pueden resolver sistemas de ecuaciones:

```
> solve({(x-1)/(3*x)<1,(abs(x)+2)/x<1},x);
```

$$\left\{x<-\frac{1}{2}\right\}$$

3) Dibujar la gráfica de la función $\sqrt{\ln\left(\frac{5\,x}{4}-\frac{x^2}{4}\right)}$ en su dominio de definición.

En primer lugar calculamos su dominio [1, 4] resolviendo una desigualdad:

```
> solve({ln(5/4*x-x^2/4)>=0},x);
```

$$\{1\le x, x\le 4\}$$

Para dibujar la curva usamos el comando **plot** (consultar en *Help on "plot"*).

```
> plot(sqrt(log((5*x-x^2)/4)),x=1..4);
```



No es una semicircunferencia (nótese que la escala es diferente en cada eje).
Se puede cambiar el tamaño del dibujo tras seleccionarlo.

4) Hallar el dominio de la función $f(x)=\frac{x^3+x^2+x+1}{x^4+x^3+3\,x^2+x+2}$ y simplificarla, si es posible.

```
> restart;
> f:=x->(x^3+x^2+x+1)/(x^4+x^3+3*x^2+x+2);
```

$$f:=x\to\frac{x^3+x^2+x+1}{x^4+x^3+3\,x^2+x+2}$$

```
> solve({x^4+3*x^2+x^3+x+2=0});
```

$$\{x=I\}, \{x=-I\}, \left\{x=-\frac{1}{2}+\frac{1}{2}\,I\,\sqrt{7}\right\}, \left\{x=-\frac{1}{2}-\frac{1}{2}\,I\,\sqrt{7}\right\}$$

Como las raíces del denominador son complejas (no reales), dicho denominador no se anula en ningún punto del eje real. Por tanto el dominio de definición de *f* es todo **R.**

```
> simplify(f(x));
```

$$\frac{x+1}{x^2+x+2}$$

5) Definir y hallar los dominios de las funciones $g(x)=\log\left(\frac{x^2-5\,x+6}{x^2+4\,x+6}\right)$ y $h(x)=\log(x-2)+\log(x-3)-\log\left(x^2+4x+6\right)$. ¿Son iguales las funciones *g* y *h*?

```
> restart;
  g:=x->log((x^2-5*x+6)/(x^2+4*x+6));
  h:=x->log(x-2)+log(x-3)-log(x^2+4*x+6);
```

$$g:=x\to\log\left(\frac{x^2-5\,x+6}{x^2+4\,x+6}\right)$$

$$h:=x\to\log(x-2)+\log(x-3)-\log\left(x^2+4\,x+6\right)$$

Veamos cuál es el dominio de la función *g*.

```
> solve({(x^2-5*x+6)/(x^2+4*x+6)>0});
```

$$\{x<2\}, \{3<x\}$$

La respuesta es suficientemente explicativa. La expresión es positiva en $(-\infty,2)\cup(3,\infty)$ y, por tanto, este es el dominio de definición de *g*. Ahora, el dominio de *h*:

```
> solve({x-2>0,x-3>0,x^2+4*x+6>0});
```

$$\{3<x\}$$

El dominio de definición de *h* es el intervalo $(3,\infty)$. Desde luego, los dominios de definición de $h(x)$ y $g(x)$ son diferentes luego, en rigor, no son funciones iguales. Una sencilla operación en el papel nos demuestra que si $3<x$, ambas funciones coinciden. Maple nos puede demostrar este hecho de la siguiente manera:

```
> simplify(g(x)-h(x));
```

$$\ln\left(\frac{(x-2)\,(x-3)}{x^2+4\,x+6}\right)-\ln(x-2)-\ln(x-3)+\ln\left(x^2+4\,x+6\right)$$

No hemos conseguido que simplifique. Precisamos algo más: queremos que considere que $3<x$; ahora el resultado sí es el que queríamos:

```
> assume(x>3):simplify(g(x)-h(x));
```

$$0$$

6) Definir la función $f(x) = \min(|x|, |x-1|)$.
A continuación, definirla sin usar órdenes que afecten a valores absolutos y mínimos.

```
> restart;f:=x->min(abs(x),abs(x-1));
```

$$f := x \to \min(|x|, |x-1|)$$

Definamos f de otra manera: si consideramos los intervalos $(-\infty, 0)$, $[0, \frac{1}{2})$, $[\frac{1}{2}, 1)$ y $[1, \infty)$, es fácil ver que la función vale $-x, x, 1-x$ y $x-1$, respectivamente. Esto lo indicamos con la orden **piecewise** de la siguiente manera (llamamos g a la nueva función):

```
> g:=x->piecewise(x<0,-x,x<1/2,x,x<1,1-x,x-1);
```

$$g := x \to piecewise\left(x < 0, -x, x < \frac{1}{2}, x, x < 1, 1-x, x-1\right)$$

Podemos ver la función escrita de manera más tradicional:

```
> g(x);
```

$$\begin{cases} -x & x < 0 \\ x & x < \frac{1}{2} \\ 1-x & x < 1 \\ -1+x & otherwise \end{cases}$$

Y ahora podemos ver que las dos funciones f y g son iguales, representando la diferencia. Para eso, utilizamos la orden **plot** e indicamos que el rango de x que queremos es todo **R**, es decir, desde $-\infty$ hasta ∞ . Iremos viendo más detalles sobre la orden **plot** en lo sucesivo.

```
> plot(f(x)-g(x),x=-infinity..infinity,thickness=3);
```



Podemos pedir a Maple que transforme la definición de la función f y nos la convierta en una expresión a trozos (no siempre será capaz de hacerlo, claro) y comprobamos que es lo mismo:

```
> convert(f(x),piecewise);
```

$$\begin{cases} -x & x < 0 \\ x & x \le \frac{1}{2} \\ 1-x & x < 1 \\ -1+x & 1 \le x \end{cases}$$

2. Gráficas de funciones

7) Dibujar la gráfica de la función $f(x) = \dfrac{x^2-1}{x^3+x-2}$ en los intervalos (-2,2) , (-100,100) y **R**.
Observar los cambios de escala que se producen.

```
> restart;
```

Por comodidad, definimos la función:

```
> f:=x->(x^2-1)/(x^3+x-2);
```

$$f := x \to \frac{x^2-1}{x^3+x-2}$$

```
> plot(f(x),x=-2..2,gridlines=true);
```




```
> plot(f(x),x=-100..100);
```




```
> plot(f(x),x=-infinity..infinity);
```



[Se ha reducido el tamaño de las 2 últimas gráficas manualmente para ajustarlas a la página].

8) Dibujar simultáneamente en el intervalo (–4,4) las gráficas de $g(x) = \log\left(\frac{x^2 - 5x + 6}{x^2 + 4x + 6}\right)$ y $h(x) = \log(x-2) + \log(x-3) - \log(x^2 + 4x + 6)$ e interpretar el resultado.

```
> restart;
> g:=x->log((x^2-5*x+6)/(x^2+4*x+6)):
  h:=x-> log(x-2)+log(x-3)-log(x^2+4*x+6):
```

Con la orden **plot**, podemos representar conjuntamente varias funciones:

```
> plot([g(x),h(x)],x=-4..4);
```



Los dos tramos de curva corresponden a la función *g*, mientras que a *h* solo corresponde el tramo entre 3 y 4. Una forma de ver las dos gráficas es dibujarlas con otro trazo. Por ejemplo, dibujemos la gráfica de *g* en verde y con puntos, y la gráfica de *h* en azul y con líneas:

```
> plot([g(x),h(x)],x=-4..4,color=[green,blue],
  style=[point,line]);
```



3. Programación

9) Crear un procedimiento para dibujar simultáneamente en el intervalo (-2,2), con distintos colores, las gráficas de una función $f(x)$ y de $f(x)+c$. Utilizarlo con algún ejemplo concreto.

```
> restart;
```

Escribamos un pequeño programa o procedimiento que toma como datos una función f y una constante c y devuelve la gráfica de las funciones f y f+c. Al procedimiento le debemos dar un nombre; por ejemplo, *graf*.

```
> graf:=proc(f::procedure,c)
  plot([f(x),f(x)+c],x=-2..2,color=[red,blue]);
  end proc;
```

graf := **proc**(*f*::*procedure*, *c*)

plot([$f(x)$, $f(x)+c$], $x=-2..2$, *color* = [*red*, *blue*])

end proc

Si queremos utilizar el procedimiento que hemos escrito, debemos definir una función antes:

```
> f:=x->x^2+1:graf(f,2);
```



En realidad, el programa es tan sencillo que podemos hacer lo mismo sin definir un procedimiento:

```
> f:=x->x^2+1:c:=2:
  plot([f(x),f(x)+c],x=-2..2,color=[red,blue]);
```



10) Crear la función de Dirichlet, que vale1 si x es racional y 0 en caso contrario.

```
> restart;
> dirichlet:=proc(x)
  if type(x,rational)=true then 1;
  else 0;
  end if;
  end proc;
```

dirichlet := **proc**(*x*) **if** *type*(*x*, *rational*) = *true* **then** 1 **else** 0 **end if end proc**

```
> dirichlet(Pi);dirichlet(32/17);dirichlet(0.345);dirichlet(a);
```

$$0$$
$$1$$
$$0$$
$$0$$

El problema está en que para Maple 0.345 no es de tipo `rational`, sino de tipo `float`, y la respuesta a la última entrada tampoco es muy correcta. Una forma más elaborada de definir esta función es la siguiente:

```
> diri:=proc()
  if nargs>1 then error "esta función requiere un solo
  argumento"
  end if;
  if type(args[1],rational) or type(args[1],float) then 1;
  elif type(args[1],realcons) then 0;
  else piecewise(args[1]=rational,1,0);
  end if
  end proc:
> diri(0.25);diri(34/27);diri(Pi);diri(exp(3));
```

$$1$$
$$1$$
$$0$$
$$0$$

```
> diri(x^3+3);
```

$$\begin{cases} 1 & x^3+3 = \mathit{rational} \\ 0 & \mathit{otherwise} \end{cases}$$